

## Programming sound and music

When you've had a little more practice with making music, then you can get a little more involved, by using the PEEK function. PEEK is a function that is equal to the value currently in the indicated memory location.

```
X=PEEK(MEM)
```

The value of the variable X is set equal to the current contents of memory location MEM. Of course, your programs include other BASIC commands, but for a full explanation of them, refer to your BASIC manual.

Let's jump right in and try a simple program using only one of the three voices. Computer ready? Then type in this program, and save it on your disk. Then, RUN it.

### EXAMPLE PROGRAM:

```
10 For L=0 TO 24:OUT &H2A,L:OUT &H2B,0:NEXT:REM CLEAR SOUND CHIP
20 OUT &H2A,5:OUT &H2B,9:OUT &H2A,6:OUT &H2B,0
30 OUT &H2A,24:OUT &H2B,15           :REM SET VOLUME TO MAXIMUM
40 READ HF,LF,DR
50 If HF<0 Then END
60 OUT &H2A,1:OUT &H2A,HF:OUT &H2A,0:OUT &H2B,LF
70 OUT &H2A,4:OUT &H2B,33
80 For T=1 TO DR:NEXT
90 OUT &H2A,4:OUT &H2B,32:For T=1 TO 50:NEXT
100 GOTO40
110 DATA25,177,250,28,214,250
120 DATA25,177,250,25,177,250
130 DATA25,177,125,28,214,125
140 DATA32,94,750,25,177,250
150 DATA28,214,250,19,63,250
160 DATA19,63,250,19,63,250
170 DATA21,154,63,24,63,63
180 DATA25,177,250,24,63,125
190 DATA19,63,250,-1,-1,-1
```

Here's a line-by-line description of the program you've just typed in. Refer to it whenever you feel the need to investigate parts of the program that you don't understand completely.

### LINE-BY-LINE EXPLANATION OF EXAMPLE PROGRAM 1:

Line(s)	Description
10	Clear all sound chip registers.
20	Set Attack/Decay for voice 1 (A=0,D=9).
30	Set Sustain/Release for voice 1 (S=0,R=0), Set volume at maximum.
40	Read high frequency, low frequency, duration of note.
50	When high frequency less than zero, song is over.
60	Poke (out on MSX) high and low frequency of voice 1.
70	Gate sawtooth waveform for voice 1.
80	Timing loop for duration of note.
90	Release sawtooth waveform for voice 1.
100	Return for next note.
110-180	Data for song: high frequency, low frequency, duration (number of counts) for each note.
190	Last note of song and negative is signaling end of song.

## Volume control

Chip register 24 contains the overall volume control. The volume can be set anywhere between 0 and 15. The other four bits are used for purposes we'll get into later. For now it is enough to know volume is 0 to 15.

Look at line 30 to see how it's set in Example Program 1.

## Frequencies of sound waves

Sound is created by the movement of air in waves. Think of throwing a stone into a pool and seeing the waves radiate outward. When similar waves are created in air, we hear it. If we measure the time between one peak of a wave and the next, we find the number of seconds for one cycle of the wave ( $n$  = number of seconds). The reciprocal of this number ( $1/n$ ) gives you the cycles per second. Cycles per second are more commonly known as the frequency. The highness or lowness of a sound (pitch) is determined by the frequency of the sound waves produced. The sound generator in your expander uses two locations to determine the frequency. Appendix E gives you the frequency values you need to reproduce a full eight octaves of musical notes. To create a frequency other than the ones listed in the note table use "Fout" (frequency output) and the following formula to represent the frequency ( $F_n$ ) of the sound you want to create. Remember that each note requires both a high and a low frequency number.

$$F_n = F_{out} / .06097$$

Once you've figured out what  $F_n$  is for your "new" note the next step is to create the high and low frequency values for that note. To do this you must first round off  $F_n$  so that any numbers to the right of the decimal point are left off. You are now left with an integer value. Now you can set the high frequency location ( $F_{hi}$ ) by using the formula  $F_{hi} = \text{INT}(F_n / 256)$  and the low frequency location ( $F_{lo}$ ) should be  $F_{lo} = F_n - (256 * F_{hi})$ .

At this point you have already played with one voice of your SID. If you wanted to stop here you could find a copy of your favorite tune and become the maestro conducting your own computer orchestra in your "at home" concert hall.

## Using multiple voices

Your Expander has three independently controlled voices (oscillators). Our first example program used only one of them. later on, you'll learn how to change the quality of the sound made by the voices. But right now, let's get all three voices singing. This example program shows you one way to translate sheet music for your computer orchestra. Try typing it in, and then SAVE it on your disk.

## EXAMPLE PROGRAM 2:

```
5 'USING MULTIPLE VOICES
10 For L=0 TO 24:OUT &H2A,L:OUT &H2B,0:NEXT
20 Dim H(2,200),L(2,200),C(2,200)
30 Dim FQ(11)
40 V(0)=17:V(1)=65:V(2)=33
50 OUT &H2A,10:OUT &H2B,8:OUT &H2A,22:OUT &H2B,128:OUT &H2A,23:OUT &H2B,244
60 For I=0 TO 11:READ FQ(I):NEXT
100 For K=0 TO 2
110 I=0
120 READ NM
130 If NM=0 Then 250
140 WA=V(K):WB=WA-1:If NM<0 Then NM=-NM:WA=0:WB=0
150 DR%NM/128:OC%=(NM-128*DR%)/16
160 NT=NM-128*DR%-16*OC%
170 FR=FQ(NT)
180 If OC%=7 Then 200
190 For J=6 TO OC% STEP-1:FR=FR/2:NEXT
200 HF%=FR/256:LF%=FR-256*HF%
210 If DR%=1 Then H(K,I)=HF%:L(K,I)=LF%:C(K,I)=WA:I=I+1:GoTo 120
220 For J=1 TO DR%-1:H(K,I)=HF%:L(K,I)=LF%:C(K,I)=WA:I=I+1:NEXT
230 H(K,I)=HF%:L(K,I)=LF%:C(K,I)=WB
240 I=I+1:GOTO120
250 If I>IM Then IM=I
260 NEXT
500 OUT &H2A,5:OUT &H2B,0:OUT &H2A,6:OUT &H2B,240
510 OUT &H2A,12:OUT &H2B,85:OUT &H2A,13:OUT &H2B,133
520 OUT &H2A,19:OUT &H2B,10:OUT &H2A,20:OUT &H2B,197
530 OUT &H2A,24:OUT &H2B,31
540 For I=0 TO IM
550 OUT &H2A,0:OUT &H2B,L(0,I):OUT &H2A,7:OUT &H2B,L(1,I):OUT &H2A,14:OUT &H2B,L(2,I)
560 OUT &H2A,1:OUT &H2B,H(0,I):OUT &H2A,8:OUT &H2B,H(1,I):OUT &H2A,15:OUT &H2B,H(2,I)
570 OUT &H2A,4:OUT &H2B,C(0,I):OUT &H2A,11:OUT &H2B,C(1,I):OUT &H2A,18:OUT &H2B,C(2,I)
580 For T=1 TO 80:NEXT:NEXT
590 For T=1 TO 200:NEXT:OUT &H2A,24:OUT &H2B,0
600 DATA 34334,36376,38539,40830
610 DATA 43258,45830,48556,51443
620 DATA 54502,57743,61176,64814
1000 DATA 594,594,594,596,596,1618,587,592,587.585,331,336
1010 DATA 1097,583,585,585,585,587,587,1609,585,331,337,594,594,593
1020 DATA 1618,594,596,594,592,587,1616,587,585,331,336,841,327
1999 DATA 1607,0
2000 DATA 583,585,583,583,327,329,1611,583,585,578,578,578
2010 DATA 196,198,583,326,578,326,327,329,327,329,326,578,583
2020 DATA 1606,582,322,324,582,587,329,327,1606,583,327,329,587,331,329
2999 DATA 329,328,1609,578,834,324,322,327,585,1602,0
3000 DATA 567,566,567,304,306,308,310,1591,567,311,310,567
3010 DATA 306,304,299,308,304,171,176,306,291,551,306,308
3020 DATA 310,308,310,306,295,297,299,304,1586,562,567,310,315,311
3030 DATA 308,313,297,1586,567,560,311,309,308,309,306,308
3999 DATA 1577,299,295,306,310,311,304,562,546,1575,0
```

Here is a line,-by-line explanation of Example Program 2. For now, we are interested in how the three voices are controlled.

#### LINE-BY-LINE EXPLANATION OF EXAMPLE PROGRAM 2:

Line(s)	Description
10	Set S equal to start of sound chip and clear all sound chip registers.
20	Dimension arrays to contain activity of song, 1/16th of a measure per location.
30	Dimension array to contain base frequency for each note.
40	Store waveform control byte for each voice.
50	Set high pulse width for voice 2. Set high frequency for filter cutoff. Set resonance for filter and filter voice 3.
60	Read in base frequency for each note.
100	Begin decoding loop for each voice.
110	Initialize pointer to activity array.
120	Read coded note.
130	If coded note is zero, then next voice.
140	Set waveform controls to proper voice. If silence, set waveform controls to 0.
150	Decode duration and octave.
160	Decode note.
170	Get base frequency for this note.
180	If highest octave, skip division loop.
190	Divide base frequency by 2 appropriate number of times.
200	Get high and low frequency bytes.
210	If sixteenth note, set activity array: high frequency, low frequency, and waveform control (voice on).
220	For all but last beat of note, set activity array: high frequency, low frequency, waveform control (voice on).
230	For last beat of note, set activity array: high frequency, low frequency, waveform control (voice off).
240	Increment pointer to activity array. Get next note.
250	If longer than before, reset number of activities.
260	Go back for next voice.
500	Set Attack/Decay for voice 1 (A=0, D=0). Set Sustain/Release for voice 1 (S=15, R=0).
510	Set Attack/Decay for voice 2 (A=5, D=5). Set Sustain/Release for voice 2 (S=8, R=5).
520	Set Attack/Decay for voice 3 (A=0, D=10). Set Sustain/Release for voice 3 (S=12, R=5).
530	Set volume 15, low-pass filtering.
540	Start loop for every 1/16th of a measure.
550	POKE low frequency from activity array for all voices.
560	POKE high frequency from activity array for all voices.
570	POKE waveform control from activity array for all voices.
580	Timing loop for 1/16th of a measure and back for next 1/16th measure.
590	Pause, then turn off volume.
600-620	Base frequency data.
1000-1999	Voice 1 data.
2000-2999	Voice 2 data.
3000-3999	Voice 3 data.

The values used in the data statements were found by using the note table in Appendix E and the chart below:

NOTE TYPE	DURATION
1/16	128
1/8	256
DOTTED 1/8	384
1/4	512
1/4+1/16	640
DOTTED 1/4	768
1/2	1024
1/2+1/16	1152
1/2+1/8	1280
DOTTED 1/2	1536
WHOLE	2048

The note number from the note table is added to the duration above. Then each note can be entered using only one number which is decoded by your program. This is only one method of coding note values. You may be able to come up with one with which you are more comfortable. The formula used here for encoding a note is as follows:

- 1) The duration (number of 1/16ths of a measure) is multiplied by 8.
- 2) The result of step 1 is added to the octave you've chosen (0-7).
- 3) The result of step 2 is then multiplied by 16.
- 4) Add your note choice (0-11) to the result of the operation in step 3.

In other words:

$$(((D*8) + O) * 16) + N$$

Where D = duration, O = octave, and N = note

A silence is obtained by using the negative of the duration number (number of 1/16ths of a measure \* 128).

## Controlling multiple voices


Once you have gotten used to using more than one voice, you will find that the timing of the three voices needs to be coordinated. This is accomplished in this program by:

- 1) Divide each musical measure into 16 parts.
- 2) Store the events that occur in each 1/16th measure interval in three separate arrays.


The high and low frequency bytes are calculated by dividing the frequencies of the highest octave by two (lines 180 and 190). The waveform control byte is a start signal for beginning a note or continuing a note that is already playing. It is a stop signal to end a note. The waveform choice is made once for each voice in line 40. Again, this is only one way to control multiple voices. You may come up with your own methods. However, you should now be able to take any piece of sheet music and figure out the notes for all three voices.

## Changing waveforms

The tonal quality of a sound is called the timbre. The timbre of a sound is determined primarily by its "waveform." If you remember the example of throwing a pebble into the water you know that the waves ripple evenly across the pond. These waves almost look like the first sound wave we're going to talk about, the sinusoidal wave, or sine wave for short (shown below).



To make what we're talking about a bit more practical, let's go back to the first example program to investigate different waveforms. The reason for this is that you can hear the changes more easily using only one voice. LOAD the first music program that you typed in earlier, from your disk, and RUN it again. That program is using the sawtooth waveform (shown here)



from the 6581 SID chip's sound generating device. Try changing the note start number in line 70 from 33 to 17 and the note stop number in line 90 from 32 to 16. Your program should now look like this:

### EXAMPLE PROGRAM 3 (EXAMPLE 1 MODIFIED):


```

10 For L=0 TO 24:OUT &H2A,L:OUT &H2B,0:NEXT
20 OUT &H2A,5:OUT &H2B,9:OUT &H2A,6:OUT &H2B,0
30 OUT &H2A,24:OUT &H2B,15
40 READ HF,LF,DR
50 If HF<0 Then END
60 OUT &H2A,1:OUT &H2B,HF:OUT &H2A,0:OUT &H2B,LF
70 OUT &H2A,4:OUT &H2B,17
80 For T=1 TO DR:NEXT
90 OUT &H2A,4:OUT &H2B,16:For T=1 TO 50:NEXT
100 GoTo 40
110 DATA 25,177,250,28,214,250
120 DATA 25,177,250,25,177,250
130 DATA 25,177,125,28,214,125
140 DATA 32,94,750,25,177,250
150 DATA 28,214,250,19,63,250
160 DATA 19,63,250,19,63,250
170 DATA 21,154,63,24,63,63
180 DATA 25,177,250,24,63,125
190 DATA 19,63,250,-1,-1,-1

```

Now RUN the program.

Notice how the sound quality is different, less twangy, more hollow. That's because we changed the sawtooth waveform into a triangular waveform (shown left). The third musical waveform is called a variable pulse wave (shown right).



PULSE WIDTH

It is a rectangular wave and you determine the length of the pulse cycle by defining the proportion of the wave which will be high. This is accomplished for voice 1 by using registers 2 and 3: Register 2 is the low byte of the pulse width (Lpw = 0 through 255). Register 3 is the high 4 bits (Hpw = 0 through 15). Together these registers specify a 12-bit number for your pulse width, which you can determine by using the following formula:

$$PWn = Hpw * 256 + Lpw$$

The pulse width is determined by the following equation:

$$PW_{out} = (PWn / 40.95) \%$$

When PWn has a value of 2048, it will give you a square wave. That means that register 2 (Lpw) = 0 and register 3 (Hpw) = 8.

Now try adding this line to your program:


```

15 OUT &H2A,3:OUT &H2B,8:OUT &H2A,2:OUT &H2B,0

```

Then change the start number in line 70 to 65 and the stop number in line 90 to 64, and RUN the program. Now change the high pulse width (register 3 in line 15) from an 8 to a 1. Notice how dramatic the difference in sound quality is?

The last waveform available to you is white noise (shown here).



It is used mostly for sound effects and such. To hear how it sounds, try changing the start number in line 70 to 129 and the stop number in line 90 to 128.

## Understanding waveforms


When a note is played, it consists of a sine wave oscillating at the fundamental frequency and the harmonics of that wave. The fundamental frequency defines the overall pitch of the note. Harmonics are sine waves having frequencies which are integer multiples of the fundamental frequency. A sound wave is the fundamental frequency and all of the harmonics it takes to make up that sound.

In musical theory let's say that the fundamental frequency is harmonic number 1. The second harmonic has a frequency twice the fundamental frequency, the third harmonic is three times the fundamental frequency, and so on. The amounts of each harmonic present in a note give it its timbre. An acoustic instrument, like a guitar or a violin, has a very complicated harmonic structure. In fact, the harmonic structure may vary as a single note is played. You have already played with the waveforms available in your SID. Now let's talk about how the harmonics work with the triangular, sawtooth, and rectangular waves. A triangular wave contains only odd harmonics. The amount of each harmonic present is proportional to the reciprocal of the square of the harmonic number. In other words harmonic number 3 is  $1/9$  quieter than harmonic number 1, because the harmonic 3 squared is 9 ( $3 \times 3$ ) and the reciprocal of 9 is  $1/9$ . As you can see, there is a similarity in shape of a triangular wave to a sine wave oscillating at the fundamental frequency. Sawtooth waves contain all the harmonics. The amount of each harmonic present is proportional to the reciprocal of the harmonic number. For example, harmonic number 2 is  $1/2$  as loud as harmonic number 1. The square wave contains odd harmonics in proportion to the reciprocal of the harmonic number. Other rectangular waves have varying harmonic content. By changing the pulse width, the timbre of the sound of a rectangular wave can be varied tremendously.

By choosing carefully the waveform used, you can start with a harmonic structure that looks somewhat like the sound you want. To refine the sound, you can add another aspect of sound quality available on your PlaySoniq called filtering, which we'll discuss later in this section.

## The envelope generator

The volume of a musical tone changes from the moment you first hear it, all the way through until it dies out and you can't hear it anymore. When a note is first struck, it rises from zero volume to its peak volume. The rate at which this happens is called the ATTACK. Then, it falls from the peak to some middle-ranged volume. The rate at which the fall of the note occurs is called the DECAY. The mid-ranged volume itself is called the SUSTAIN level. And finally, when the note stops playing, it falls from the SUSTAIN level to zero volume. The rate at which it falls is called the RELEASE. Here is a sketch of the four phases of a note:



Each of the items mentioned above give certain qualities and restrictions to a note. The bounds are called parameters. The parameters ATTACK/DECAY/SUSTAIN/RELEASE and collectively called ADSR, can be controlled by your use of another set of locations in the sound generator chip. LOAD your first example program again. RUN it again and remember how it sounds. Then, changing line 20 so the program is like this:

### EXAMPLE PROGRAM 4 (EXAMPLE 1 MODIFIED):

```
10 For L=0 TO 24:OUT &H2A,L:OUT &H2B,0:NEXT
20 OUT &H2A,5:OUT &H2B,8:OUT &H2A,6:OUT &H2B,195
30 OUT &H2A,24:OUT &H2B,15
40 READ HF,LF,DR
50 If HF<0 Then END
60 OUT &H2A,1:OUT &H2B,HF:OUT &H2A,0:OUT &H2B,LF
70 OUT &H2A,4,:OUT &H2B,33
80 For T=1 TO DR:NEXT
90 OUT &H2A,4:OUT &H2B,32:For T=1 TO 50:NEXT
100 GoTo 40
110 DATA 25,177,250,28,214,250
120 DATA 25,177,250,25,177,250
130 DATA 25,177,125,28,214,125
140 DATA 32,94,750,25,177,250
150 DATA 28,214,250,19,63,250
160 DATA 19,63,250,19,63,250
170 DATA 21,154,63,24,63,63
180 DATA 25,177,250,24,63,125
190 DATA 19,63,250,-1,-1,-1
```

Registers 5 and 6 define the ADSR for voice 1. The ATTACK is the high nibble of register 5. A Nibble is half a byte, in other words the lower 4 or higher 4 on/off locations (bits) in each register. DECAY is the low nibble. You can pick any number 0 through 15 for ATTACK, multiply it by 16 and add to any number 0 through 15 for DECAY. The values that correspond to these numbers are listed below.

SUSTAIN level is the high nibble of register 6. It can be 0 through 15. It defines the proportion of the peak volume that the SUSTAIN level will be. RELEASE rate is the low nibble of register 6.

Here are the meanings of the values for ATTACK, DECAY, and RELEASE:

VALUE	ATTACK RATE (TIME/CYCLE)	DECAY/RELEASE RATE (TIME/CYCLE)
0	2 ms	6 ms
1	8 ms	24 ms
2	16 ms	48 ms
3	24 ms	72 ms
4	38 ms	114 ms
5	56 ms	168 ms
6	68 ms	204 ms
7	80 ms	240 ms
8	100 ms	300 ms
9	250 ms	750 ms
10	500 ms	1.5 s
11	800 ms	2.4 s
12	1 s	3 s
13	3 s	9 s
14	5 s	15 s
15	8 s	24 s

Here are a few sample settings to try in your example program. Try these and a few of your own. The variety of sounds you can produce is astounding! For a violin type sound, try changing line 20 to read:

```
20 OUT &H2A,5:OUT &H2B,8:OUT &H2A,6:OUT &H2B,89: :REM A=5;D=8;S=5;R=9
```

Change the waveform to triangle and get a xylophone type sound by using these lines:

```
20 OUT &H2A,5:OUT &H2B,9:OUT &H2A,6:OUT &H2B,9:REM A=0;D=9;S=0;R=9
70 OUT &H2A,4,:OUT &H2B,17
90 OUT &H2A,4,:OUT &H2B,16:FORT=1TO50:NEXT
```

Change the waveform to square and try a piano type sound with these lines:

```
15 OUT &H2A,3:OUT &H2B,8:OUT &H2A,2:OUT &H2B,0
20 OUT &H2A,5:OUT &H2B,9:OUT &H2A,6:OUT &H2B,0: REM A=0;D=9;S=0;R=0
70 OUT &H2A,4:OUT &H2B,65
90 OUT &H2A,4:OUT &H2B,64: :FORT=1TO50:NEXT
```

The most exciting sounds are those unique to the music synthesizer itself, ones that do not attempt to mimic acoustic instruments. For example try:

```
20 OUT &H2A,5:OUT &H2B,144:OUT &H2A,6:OUT &H2B,243:REM A=9;D=0; S=15;R=3
```

## Filtering

The harmonic content of a waveform can be changed by using a filter. The SID chip is equipped with three types of filtering. They can be used separately or in combination with one another. Let's go back to the sample program you've been using to play with a simple example that uses a filter. There are several filter controls to set. You add line 15 in the program to set the cutoff frequency of the filter. The cutoff frequency is the reference point for the filter. You SET the high and low frequency cutoff points in registers 21 and 22. To turn ON the filter for voice 1, POKE register 23. Next change line 30 to show that a high-pass filter will be used (see the SID register map).


### EXAMPLE PROGRAM 5 (EXAMPLE 1 MODIFIED):

```


10 For L=0 TO 24:OUT &H2A,L:OUT &H2B,0:NEXT:REM CLEAR SOUND CHIP
15 OUT &H2A,22:OUT &H2B,28:OUT &H2A,21:OUT &H2B,0:OUT &H2A,23:OUT &H2B,1
20 OUT &H2A,5:OUT &H2B,9:OUT &H2A,6:OUT &H2B,0:
30 OUT &H2A,24:OUT &H2B,79
40 READ HF,LF,DR
50 If HF<0 Then END
60 OUT &H2A,1:OUT &H2A,HF:OUT &H2A,0:OUT &H2B,LF
70 OUT &H2A,4:OUT &H2B,33
80 For T=1 TO DR:NEXT
90 OUT &H2A,4:OUT &H2B,32:For T=1 TO 50:NEXT
100 GOTO40
110 DATA25,177,250,28,214,250
120 DATA25,177,250,25,177,250
130 DATA25,177,125,28,214,125
140 DATA32,94,750,25,177,250
150 DATA28,214,250,19,63,250
160 DATA19,63,250,19,63,250
170 DATA21,154,63,24,63,63
180 DATA25,177,250,24,63,125
190 DATA19,63,250,-1,-1,-1

```


Try running the program now. Notice the lower tones have had their volume cut down. It makes the overall quality of the note sound tinny. This is because you are using a high-pass filter which attenuates (cuts down the level of) frequencies below the specified cutoff frequency. There are three types of filters in your Expander's SID chip. We have been using the high-pass filter. It will pass all the frequencies at or above the cutoff, while attenuating the frequencies below the cutoff.




The SID chip also has a low-pass filter. As its name implies, this filter will pass the frequencies below cutoff and attenuate those above.



Finally, the chip is equipped with a bandpass filter, which passes a narrow band of frequencies around the cutoff, and attenuates all others.



The high- and low-pass filters can be combined to form a notch reject filter which passes frequencies away from the cutoff while attenuating at the cutoff frequency.



Register 24 determines which type filter you want to use. This is in addition to register 24's function as the overall volume control. Bit 6 controls the high-pass filter (0 = off, 1 = on), bit 5 is the bandpass filter, and bit 4 is the low-pass filter. The low 3 bits of the cutoff frequency are determined by register 21 (Lcf) (Lcf = 0 through 7). While the 8 bits of the high cutoff frequency are determined by register 22 (Hcf) (Hcf = 0 through 255). Through careful use of filtering, you can change the harmonic structure of any waveform to get just the sound you want. In addition, changing the filtering of a sound as it goes through the ADSR phases of its life can produce interesting effects.

### Advanced techniques

The SID chip's parameters can be changed dynamically during a note or sound to create many interesting and fun effects. In order to make this easy to do, digitized outputs from oscillator three and envelope generator three are available for you in registers 27 and 28, respectively. The output of oscillator 3 (register 27) is directly related to the waveform selected. If you choose the sawtooth waveform of oscillator 3, this register will present a series of numbers incremented (increased step by step) from 0 to 255 at a rate determined by the frequency of oscillator 3. If you choose the triangle waveform, the output will increment from 0 up to 255, then decrement (decrease step by step) back down to 0. If you choose the pulse wave, the output will jump back-and-forth between 0 and 255. Finally, choosing the noise waveform will give you a series of random numbers. When oscillator 3 is used for modulation, you usually do NOT want to hear its output. Setting bit 7 of register 24 turns the audio output of voice 3 off. Register 27 always reflects the changing output of the oscillator and is not affected in any way by the envelope (ADSR) generator.

Register 25 gives you access to the output of the envelope generator of oscillator 3. It functions in much the same fashion that the output of oscillator 3 does. The oscillator must be turned on to produce any output from this register. Vibrato (a rapid variation in frequency) can be achieved by adding the output of oscillator 3 to the frequency of another oscillator.

Example Program 6 illustrates this idea.

## EXAMPLE PROGRAM 6:

```
20 For L=0 TO 24:OUT &H2A,L:OUT &H2B,0:NEXT:REM CLEAR SOUND CHIP
30 OUT &H2A,3:OUT &H2B,8
40 OUT &H2A,5:OUT &H2B,41:OUT &H2A,6:OUT &H2B,89
50 OUT &H2A,14:OUT &H2B,117
60 OUT &H2A,18:OUT &H2B,16
70 OUT &H2A,24:OUT &H2B,143
80 READ FR,DR
90 If FR=0 Then END
100 OUT &H2A,4:OUT &H2B,65
110 For T=1 TO DR*2
120 OUT &H2A,27:FQ=FR+INP(&H2B)/2
130 HF=INT(FQ/256):LF=LQ AND 255
140 OUT &H2A,0:OUT &H2B,LF:OUT &H2A,1:OUT &H2B,HF
150 NEXT
160 OUT &H2A,4:OUT &H2B,64
170 GoTo 80
500 DATA 4817,2,5103,2,5407,2
510 DATA 8583,4,5407,2,8583,4
520 DATA 5407,4,8583,12,9634,2
530 DATA 10207,2,10814,2,8583,2
540 DATA 9634,4,10814,2,8583,2
550 DATA 8583,12
560 DATA 0,0
```

## LINE-BY-LINE EXPLANATION OF EXAMPLE PROGRAM 6:

Lines(s)	Description
20	Clear all sound chip locations.
30	Set high pulse width for voice 1.
40	Set Attack/Decay for voice 1 (A=2, D=9).
	Set Sustain/Release for voice 1 (S=5, R=9).
50	Set low frequency for voice 3.
60	Set triangle waveform for voice 3.
70	Set volume 15, turn off audio output of voice 3.
80	Read frequency and duration of note.
90	If frequency equals zero, stop.
100	POKE start pulse waveform control voice 1.
110	Start timing loop for duration.
120	Get new frequency using oscillator 3 output.
130	Get high and low frequency.
140	POKE high and low frequency for voice 1.
150	End of timing loop.
160	POKE stop pulse waveform control voice 1.
170	Go back for next note.
500-550	Frequencies and durations for song,
560	Zeros signal end of song.

A wide variety of sound effects can also be achieved using dynamic effects. For example, the following siren program dynamically changes the frequency output of oscillator 1 when it's based on the output of oscillator 3's triangular wave:

### EXAMPLE PROGRAM 7:

```
10 For L=0 TO 24:OUT &H2A,L:OUT &H2B,0:NEXT:REM CLEAR SOUND CHIP
30 OUT &H2A,14:OUT &H2B,5
40 OUT &H2A,18:OUT &H2B,16
50 OUT &H2A,3:OUT &H2B,1
60 OUT &H2A,24:OUT &H2B,143
70 OUT &H2A,6:OUT &H2B,240
80 OUT &H2A,4:OUT &H2B,65
90 FR=5389
100 For T=1 TO 200
110 OUT &H2A,27:FQ=FR+INP(&H2B)*3.5
120 HF=INT(FQ/256):LF=FQ-HF*256
130 OUT &H2A,0:OUT &H2B,LF:OUT &H2A,1:OUT &H2B,HF
140 NEXT
150 OUT &H2A,24:OUT &H2B,0
```

### LINE-BY-LINE EXPLANATION OF EXAMPLE PROGRAM 7:

Line(s)	Description
10	Set S to start of sound chip.
20	Clear sound chip registers.
30	Set low frequency of voice 3.
40	Set triangular waveform voice 3.
50	Set high pulse width for voice 1.
60	Set volume 15, turn off audio output of voice 3.
70	Set Sustain/Release for voice I (S=15, R=0).
80	POKE start pulse waveform control voice 1.
90	Set lowest frequency for siren.
100	Begin timing loop.
110	Get new frequency using output of oscillator 3.
120	Get high and low frequencies.
130	POKE high and low frequencies for voice 1.
140	End timing loop.
150	Turn off volume.

The noise waveform can be used to provide a wide range of sound effects. This example mimics a hand clap using a filtered noise waveform:

### EXAMPLE PROGRAM 8:

```
10 For L=0 TO 24:OUT &H2A,L:OUT &H2B,0:NEXT:REM CLEAR SOUND CHIP
30 OUT &H2A,0:OUT &H2B,240:OUT &H2A,1:OUT &H2B,33
40 OUT &H2A,5:OUT &H2B,8
50 OUT &H2A,22:OUT &H2B,104
60 OUT &H2A,23:OUT &H2B,1
70 OUT &H2A,24:OUT &H2B,79
80 For N=1 TO 15
90 OUT &H2A,4:OUT &H2B,129
100 For T=1 TO 250:NEXT:OUT &H2A,4:OUT &H2B,128
110 For T=1 TO 30:NEXT:NEXT
120 OUT &H2A,24:OUT &H2B,0
```

## LINE-BY-LINE EXPLANATION OF EXAMPLE PROGRAM 8:

Line(s)	Description
20	Clear all sound chip registers.
30	Set high and low frequencies for voice 1.
40	Set Attack/Decay for voice 1 (A=0, D=8).
50	Set high cutoff frequency for filter.
60	Turn on filter for voice 1.
70	Set volume 15, high-pass filter.
80	Count 15 claps.
90	Set start noise waveform control.
100	Wait, then set stop noise waveform control.
110	Wait, then start next clap-
120	Turn off volume.

## Synchronization and ring modulation

The 6581 SID chip lets you create more complex harmonic structures through synchronization or ring modulation of two voices. The process of synchronization is basically a logical ANDing of two wave forms. When either is zero, the output is zero. The following example uses this process to create an imitation of a mosquito:

```
10 'SYNCHRONIZATION AND RING MODULATION
20 For L=0 TO 24:OUT &H2A,L:OUT &H2B,0:NEXT:REM CLEAR SOUND CHIP
30 OUT &H2A,1:OUT &H2B,100
40 OUT &H2A,5:OUT &H2B,219
50 OUT &H2A,15:OUT &H2B,28
60 OUT &H2A,24:OUT &H2B,15
70 OUT &H2A,4:OUT &H2B,19
80 For T=1 TO 5000:NEXT
90 OUT &H2A,4:OUT &H2B,18
100 For T=1 TO 1000:NEXT:OUT &H2A,24:OUT &H2B,0
```

## LINE-BY-LINE EXPLANATION OF EXAMPLE PROGRAM 9:

Line(s)	Description
20	Clear sound chip registers.
30	Set high frequency voice 1.
40	Set Attack/Decay for voice 1 (A=13, D=11).
50	Set high frequency voice 3.
60	Set volume 15.
70	Set start triangle, sync waveform control for voice 1.
80	Timing loop.
90	Set stop triangle, sync waveform control for voice 1.
100	Wait, then turn off volume.

The synchronization feature is enabled (turned on) in line 70, where bits 0, 1, and 4 of register 4 are set. Bit 1 enables the syncing function between voice 1 and voice 3. Bits 0 and 4 have their usual functions of gating voice 1 and setting the triangular waveform.

Ring modulation (accomplished for voice 1 by setting bit 3 of register 4 in line 70 of the program below) replaces the triangular output of oscillator 1 with a "ring modulated" combination of oscillators 1 and 3. This produces non-harmonic overtone structures for use in mimicking bell or gong sounds. This program produces a clock chime imitation:

```

20 FOR l= 0 TO 24: OUT &H2A,I:OUT &H2B,0:NEXT
30 OUT &H2A,1:OUT &H2B,130
40 OUT &H2A,5:OUT &H2B,9
50 OUT &H2A,15:OUT &H2B,30
60 OUT &H2A,24:OUT &H2B,15
70 FOR l=1 TO 12: OUT &H2A,4:OUT &H2B,21
80 FOR T= 1 TO 1000:NEXT: OUT &H2A,4:OUT &H2B,20
90 FOR T = 1 1000:NEXT:NEXT

```

#### LINE-BY-LINE EXPLANATION OF EXAMPLE PROGRAM 10:

Line(s)	Description
20	Clear sound chip registers.
30	Set high frequency for voice 1.
40	Set Attack/Decay for voice 1 (A=0, D=9).
50	Set high frequency for voice 3.
60	Set volume 15.
70	Count number of clings, set start triangle, ring mod waveform control voice 1.
80	Timing loop, set stop triangle, ring mod.
90	Timing loop, next ding.

The effects available through the use of the parameters of your SID chip are numerous and varied. Only through experimentation on your own will you fully appreciate the capabilities of your expander. The examples in this section merely scratch the surface.